مكتبة

”خُذْ وأعطِي“

الإلكترونية

جامعة آل البيت " كلية الإقتصاد "

مجموعة طلابية تسعى لتوفير كل ما يلزم طلاب
كلية إدارة المال والاعمال من مواد وشروحات واسئلة بصورة الكترونية



Principles of Finance with Excel, 2nd edition

Instructor materials

Chapter 5
Issues in capital budgeting



Chapter 5

- Problems using IRR as decision criterion
- Choosing between projects with different lifetimes
- Mid-year discounting
- Taxation and lease/purchase
- Inflation considerations

IRR as a decision criterion

- Good points
  - IRR is simple to use
  - IRR gives information investors want
    - What is the rate of return on an investment?
- Bad points
  - IRR can represent both the *rate of return* and the *cost* of an investment
    - You can't tell without more information
  - A project can have multiple IRRs

IRR's good points

- See Chapter 3 for many examples



IRR: An investor who pays $500 to invest in this project will earn a compound annual return of 19.47%

From Chapter 3

- IRR is the rate that makes NPV = 0

	A	B	C
14	**Discount rate**	**NPV**	
15	0%	260.00	<-- =$B$4+NPV(A15,$B$5:$B$9)
16	2%	223.29	<-- =$B$4+NPV(A16,$B$5:$B$9)
17	4%	189.52	<-- =$B$4+NPV(A17,$B$5:$B$9)
18	6%	158.39	
19	8%	129.63	
20	10%	102.99	
21	12%	78.27	
22	14%	55.28	
23	16%	33.85	
24	18%	13.86	
25	20%	-4.84	
26	22%	-22.35	
27	24%	-38.78	
28			
29			
30			
31			
32			
33			
34			
35			



Problem: IRR does not distinguish between return and cost

EXAMPLE: You're buying a car that costs $11,000. Dealer offers you two payment options

- You can pay the dealer cash and get a $1,000 discount, thus paying only $10,000.
- You can pay $5,000 now and pay $2,000 in each of the next 3 years. The dealer calls this his "zero-interest car loan" plan.
- The bank is giving car loans at 9% interest, so the dealer claims that his "zero interest" plan is much cheaper.

Your cash flows and IRR

	A	B	C	D	E
1	**BUYING A CAR**				
2	List price of car	11,000.00			
3	Downpayment	5,000.00			
4	Cash cost of car	10,000.00			
5	Bank rate of interest	9%			
6					
7	**Year**	**Payment in cash**	**Payment with credit**	**Cash spent or saved with credit plan**	
8	0	-10,000.00	-5,000.00	5,000.00	<-- =C8-B8
9	1		-2,000.00	-2,000.00	<-- =C9-B9
10	2		-2,000.00	-2,000.00	
11	3		-2,000.00	-2,000.00	
12					
13	Internal rate of return			9.70%	<-- =IRR(D8:D11)

The "zero interest loan" from dealer saves you $5,000 today and costs you $2,000 in years 1, 2, 3. So it's like a loan of $5,000 with payments over three years.
The IRR of this loan is 9.70%.

More expensive than bank loan of 9%!

Better to buy car for cash and take bank loan

	A	B	C	D	E
18	**Borrowing the money from the bank**				
19	**Year**	**Payment in cash**	**Bank loan cash flows**	**Total cash flow to car owner**	
20	0	-10,000.00	5,000.00	-5,000.00	
21	1		-1,975.27	-1,975.27	<-- =PMT(9%,3,C20)
22	2		-1,975.27	-1,975.27	
23	3		-1,975.27	-1,975.27	

If you borrowed $5,000 from the bank at 9% over 3 years, annual repayment would be $1,975.27.

Car example: summary

- Dealer's "zero interest loan" really costs 9.70%.
- You should prefer the bank loan.

IRR problem: The dealer's cash flows also have IRR = 9.70%!

- This means that for the dealer, the "zero interest loan" is a good deal!

	A	B	C	D	E
1	**IRR VERSUS NPV--THE DEALER'S PROBLEM**				
2	List price of car	11,000.00			
3	Downpayment	5,000.00			
4	Cash cost of car	10,000.00			
5	Bank rate of interest	9%			
6					
7	**Year**	**Payment in cash**	**Payment with credit**	**Differential dealer cash flow**	
8	0	10,000.00	5,000.00	-5,000.00	<-- =C8-B8
9	1		2,000.00	2,000.00	<-- =C9-B9
10	2		2,000.00	2,000.00	
11	3		2,000.00	2,000.00	
12					
13	Internal rate of return			9.70%	<-- =IRR(D8:D11)

The dealer's cash flows are the same as the purchaser's cash flows, with reversed sign. Dealer also has an IRR of 9.70%!

Another IRR problem: Multiple IRRs

- You're contemplating starting a garbage dump ("Sanitary Landfill")
- Cost today: $800k
- Cash flows in years 1-5: $450k
- Cost in year 7: $1,500k
  - The cost of sanitizing after you fill up the garbage dump

	A	B	C	D
	SANITARY LANDFILL, INC.			
2	**Year**	**Cash flow**		
3	0	-800,000		
4	1	450,000		
5	2	450,000		
6	3	450,000		
7	4	450,000		
8	5	450,000		
9	6	-1,500,000		

The landfill has <u>two</u> IRR's! One is approximately 3% and the other is around 28%.

	E	F	G
2	**Discount rate**	**NPV**	
3	0%	-50,000	<-- =NPV(E3,$B$4:$B$9)+$B$3
4	2%	-10,900	<-- =NPV(E4,$B$4:$B$9)+$B$3
5	4%	17,848	<-- =NPV(E5,$B$4:$B$9)+$B$3
6	6%	38,123	
7	8%	51,465	
8	10%	59,143	
9	12%	62,203	
10	14%	61,507	
11	16%	57,769	
12	18%	51,580	
13	20%	43,428	
14	22%	33,721	
15	24%	22,793	
16	26%	10,923	
17	28%	-1,658	
18	30%	-14,758	
19	32%	-28,219	
20	34%	-41,912	
21	36%	-55,727	



Using Excel **IRR** to find both IRR's

	A	B	C	D
1	**SANITARY LANDFILL, INC.**			
2	**Year**	**Cash flow**		
3	0	-800,000		
4	1	450,000		
5	2	450,000		
6	3	450,000		
7	4	450,000		
8	5	450,000		
9	6	-1,500,000		
10				
11	First IRR	2.68%	<-- =IRR(B3:B9,0)	
12	Second IRR	27.74%	<-- =IRR(B3:B9,25%)	



Two IRRs: conclusion

- If: the correct discount rate (meaning: appropriate to project risk) is > 2.68% and < 27.74%
  - Project is a good one
- If: the correct discount rate (meaning: appropriate to project risk) is < 2.68% or > 27.74%
  - Project is not good
- Upshot: In this case you need NPV!

When do multiple IRRs occur?

- Project has “conventional cash flows” if cash flows change sign only once:
  - Initial cash flow is negative
  - All other cash flows are non-negative

  OR
  - Initial cash flow is positive
  - All other cash flows are non-positive
- Multiple IRRs can occur if project has <u>non-conventional cash flows</u>.

	A	B	C	D	E	F	G
1	**CONVENTIONAL AND NONCONVENTIONAL CASH FLOW PATTERNS**						
2	**Year**	**Cash flow Project A**	**Cash flow Project B**	**Cash flow Project C**	**Cash flow Project D**	**Cash flow Project E**	**Cash flow Project F**
3	0	-100	-100	100	25	-25	-250
4	1	200	-50	55	35	80	35
5	2	500	60	35	-200	-100	145
6	3	50	80	50	33	200	330
7	4	60	99	-100	55	55	55
8	5	35	100	-35	155	-250	-250
9		↑ Conventional cash flow pattern	↑ Conventional cash flow pattern	↑ Conventional cash flow pattern	↑ Nonconventional cash flow pattern	↑ Nonconventional cash flow pattern	↑ Nonconventional cash flow pattern
10		Initial negative cash flow followed by positive cash flows	Two initial negative cash flows followed by positive cash flows	Initial positive cash flows followed by negative cash flows	Two positive cash flows, then negative, then three positive cash flows	Initial negative cash flow, then positive, then negative, positive, negative cash flows	Negative cash flows at beginning and end, other cash flows positive

Cash flows that have multiple sign changes can have multiple IRRs!

Capital budgeting: Comparing projects with different life spans

- You're considering buying one of two trucks
- Truck A: Cheaper to buy, longer life, lower annual cash flows
- Truck B: More expensive, shorter life, but higher annual cash flows
- How to compare??

This is WRONG!

	A	B	C	D
1	**DIFFERENT LIFE SPANS**			
2	Discount rate	12%		
3				
4	**Year**	**Truck A**	**Truck B**	
5	0	-100	-250	
6	1	150	300	
7	2	150	300	
8	3	150	300	
9	4	150		
10	5	150		
11	6	150		
12				
13	NPV	516.71	470.55	<-- =C5+NPV($B$2,C6:C11)

The NPV of A is higher than B, but they're not really comparable!

One solution

	A	B	C	D
1	**DIFFERENT LIFE SPANS at end of year 3, truck B is replaced**			
2	Discount rate	12%		
3				
4	**Year**	**Cash flow (A)**	**Cash flow (B)**	
5	0	-100	-250	
6	1	150	300	
7	2	150	300	
8	3	150	50	<-- =300-250
9	4	150	300	
10	5	150	300	
11	6	150	300	
12				
13	NPV	516.71	805.48	<-- =C5+NPV($B$2,C6:C11)

Assume that after 3 years, a new truck B is purchased to replace old truck.

Now both projects are comparable.

Conclusion:
B is better than A

Second solution: Equivalent Annuity Cash Flow (EAC)

- Compute annual CF over life of project that gives same NPV as project

$$Truck\ A\ NPV = -100 + \sum_{t=1}^{6} \frac{150}{(1.12)^t} = 516.71$$

$$= \sum_{t=1}^{6} \frac{125.68}{(1.12)^t}$$

125.68 is the EAC of Truck A

Buying truck A is like getting 125.68 per year for the life of the truck.

$$Truck\ B\ NPV = -250 + \sum_{t=1}^{3} \frac{300}{(1.12)^t} = 470.55$$

$$= \sum_{t=1}^{3} \frac{195.91}{(1.12)^t}$$

195.91 is the EAC of Truck B

Buying truck B is like getting 195.91 per year for the life of the truck.

Truck A has lower EAC than truck B
Therefore B is preferred.

In Chapter 5: Tradeoff between regular light bulb and energy-saver

Standard incandescent bulb—cheap to buy, expensive to operate, short life.

Energy-saving fluorescent bulb—expensive to buy, cheap to operate, long life.

- Regular light bulb
  - Cheap to buy
  - Expensive to operate
  - Short life
- Energy-saver
  - Expensive
  - Cheap to operate
  - Long life

Chapter 5 uses EAC to compute which light is preferable.

	A	B	C
1	**LIGHT BULBS** **Choosing between cheap incandescents and expensive fluorescents**		
2	Annual discount rate	8%	
3	Monthly discount rate	0.643%	<-- =(1+B2)^(1/12)-1
4	Electric cost per kilowatt (a kilowatt = 1000 watts)	0.10	
5			
6	**Incandescent bulb**		
7	Watts	100	
8	Cost	$1.00	
9	Hours per month used	250	
10	Lifetime of bulb (hours)	1,000	
11	Lifetime in months	4	
12	Monthly cost	2.50	<-- =B9*$B$4*B7/1000
13	NPV of lifetime use	10.84	<-- =B8+PV(B3,B11,-B12)
14	Monthly equivalent annuity cash flow (EAC) for cheap incandescent	2.75	<-- =-PMT(B3,B11,B13)
15			
16	**Equivalent fluorescent bulb**		
17	Watts	15	
18	Cost	$5.00	
19	Hours per month used	250	
20	Lifetime of bulb (hours)	10,000	
21	Lifetime in months	40	
22	Monthly cost	0.38	<-- =B19*$B$4*B17/1000
23	NPV of lifetime use	18.19	<-- =B18+PV(B3,B21,-B22)
24	Monthly equivalent annuity cash flow (EAC) for expensive fluorescent	0.52	<-- =-PMT(B3,B21,B23)

Based on various costs, the conclusion of the spreadsheet:

- The monthly equivalent annuity cash flow (EAC) for the fluorescent bulb is $0.52
- The EAC for the regular bulb is $2.75

Conclusion: Cheaper to invest a lot of money in an expensive fluorescent.

Lease/Purchase with taxes

- Tax rate = 40%
- Purchase cost of computer: $4,000
  - Depreciation: over 3 years
  - $1,333/year—expense for taxes
- Lease cost: $1,500
  - Paid in advance, years 0,1, 2, 3
  - Expense for taxes
- Bank lending rate: 15%

Solution: Compare IRR of after-tax lease savings (row 16) to after-tax bank rate

	A	B	C	D	E	F
1	**LEASE OR PURCHASE?** **Costs are negative numbers and inflows positive numbers**					
2	Asset cost	4,000.00				
3	Annual depreciation if asset is purchased	1,333.33	<-- =B2/3			
4	Annual lease payment	1,500.00				
5	Bank rate	15%				
6	Tax rate	40%				
7						
8	**Year**	**0**	**1**	**2**	**3**	
9	**Purchase cash flows**					
10	Cost of machine	-4,000				
11	Depreciation tax shield		533	533	533	<-- =$B$3*$B$6
12	Total	-4,000	533	533	533	<-- =E11+E10
13						
14	**After-tax lease payments**	-900	-900	-900	-900	<-- =-$B$4*(1-$B$6)
15						
16	The lease saves	3,100	-1,433	-1,433	-1,433	<-- =-E12+E14
17						
18	IRR of lease savings	18.33%	<-- =IRR(B16:E16)			
19	Alternative cost (after-tax bank interest)	9.00%	<-- =B5*(1-$B$6)			
20						
21	Lease or purchase?	buy	<-- =IF(B18>B19,"buy","lease")			

Inflation-adjusted discounting

- Point 1: Inflation: money prices rise, purchasing power ↓
- Point 2: Anticipated future inflation makes interest rates and discount rates ↑

	A	B	C	D	E	F	G	H	I
1	**COMPUTING THE INFLATION RATE FROM THE CONSUMER PRICE INDEX (CPI)**								
2	**Year**	**U.S. consumer price index**	**Annual inflation rate**						
3	1980	82.400							
4	1981	90.900	10.32%	<-- =B4/B3-1					
5	1982	96.500	6.16%	<-- =B5/B4-1					
6	1983	99.600	3.21%						
7	1984	103.900	4.32%						
8	1985	107.600	3.56%						
9	1986	109.600	1.86%						
10	1987	113.600	3.65%						
11	1988	118.300	4.14%						
12	1989	124.000	4.82%						
13	1990	130.700	5.40%						
14	1991	136.200	4.21%						
15	1992	140.300	3.01%						
16	1993	144.500	2.99%						
17	1994	148.200	2.56%						
18	1995	152.400	2.83%						
19	1996	156.900	2.95%						
20	1997	160.500	2.29%						
21	1998	163.000	1.56%						
22	1999	166.600	2.21%						
23	2000	172.200	3.36%						
24	2001	177.100	2.85%						
25	2002	179.900	1.58%						
26	2003	184.000	2.28%						
27	2004	188.900	2.66%						
28	2005	195.300	3.39%						
29	2006	201.600	3.23%						
30	2007	207.342	2.85%						
31	2008	215.303	3.84%	<-- =B31/B30-1					
32	2009	215.522	0.10%	<-- =B32/B31-1					
33	Average annual inflation		3.37%	<-- =(B32/B3)^(1/29)-1					



Inflation adds up!

- 3% inflation per year over 10 years
- ➔ 34.39% cumulative inflation over 10 years
- ➔ $1 at end of 10 years worth only $0.7441 in purchasing power of $1 at beginning of decade

	A	B	C
1	**ANNUAL INFLATION RATES AND CUMULATIVE INFLATION**		
2	Annual inflation rate	3%	
3	Cumulative inflation over 10 years	34.39%	<-- =(1+B2)^10-1
4	End-decade $ worth in terms of beginning of decade $	0.7441	<-- =1/(1+B2)^10

	A	B	C	D
1	**WHAT'S A DOLLAR WORTH?**			
2	**Annual inflation rate**	**End-decade $ worth in terms of beginning of decade $**	**Cumulative inflation over 10 years**	
3	0%	1.00	0.00%	
4	1%	0.91	10.46%	
5	2%	0.82	21.90%	
6	3%	0.74	34.39%	
7	4%	0.68	48.02%	
8	5%	0.61	62.89%	=1/(1+A13)^10
9	6%	0.56	79.08%	
10	7%	0.51	96.72%	
11	8%	0.46	115.89%	
12	9%	0.42	136.74%	
13	10%	0.39	159.37%	<-- =(1+A13)^10-1
14				



Inflation terminology

- "Nominal cash flows": Cash flows in dollars at the time received
  - Sometimes called "current dollars"
  - Example: You are promised $100 in 3 years. No relation between the amount you will get and the inflation rate. Then $100 is the promised nominal payment.

Inflation terminology (2)

- "Real cash flows": Cash flows adjusted for changes in purchasing power.
  - You are promised $100 in 3 years. You anticipate 4% inflation per year. Then the real anticipated cash flow = $100/(1.04)^3 = $88.90.
  - $88.90 is the purchasing-power adjusted cash flow in year 0 (today's) dollars

Inflation terminology (3)

- "Nominal interest rates": Interest rates quoted today for future nominal payments.
  - This is the usual method
  - Example: You borrow $1,000 today for 1 year at 7%. Your nominal repayment in 1 year is $1000*1.07=$1,070 . This payment is unrelated to the interest rate.
  - 7% is the nominal interest rate . It applies no matter what the inflation rate is.

Inflation terminology (4)

- "Real interest rate": Interest rate adjusted for changes in purchasing power.
  - Example: You borrow $1,000 for 1 year at 7% (nominal interest rate). You anticipate inflation of 3%. Then the inflation-adjusted repayment in 1 year = $1070/1.03 = $1038.84 .
  - The real interest rate that you will pay is $1038.84/1000 -1 = 3.884%.
  - Note: 3.884% = 1.07/1.03 -1

Translating nominal cash flows to real using the CPI

	A	B	C	D	E	F	G
1	**HOW MUCH DID YOU REALLY EARN?**						
2	**Year**	**Nominal cash flow**	**CPI**	**Cumulative inflation rate**		**Real cash flow**	**<-- This is the cash flow in 1995 dollars**
3	1995	-1,000	133			-1,000.00	
4	1996	150	138	3.76%	<-- =C4/$C$3-1	144.57	<-- =B4/(1+D4)
5	1997	150	142	6.77%	<-- =C5/$C$3-1	140.49	<-- =B5/(1+D5)
6	1998	150	145	9.02%	<-- =C6/$C$3-1	137.59	
7	1999	150	148	11.28%		134.80	
8	2000	150	153	15.04%		130.39	
9	2001	150	166	24.81%		120.18	
10	2002	150	172	29.32%		115.99	
11	2003	150	180	35.34%		110.83	
12	2004	150	191	43.61%		104.45	
13	2005	1,150	195	46.62%	<-- =C13/$C$3-1	784.36	
14							
15	Nominal IRR	15.00%	<-- =IRR(B3:B13)		Real IRR	10.93%	<-- =IRR(F3:F14)

Nominal and real discount rates

- Relation between nominal and real discount rates:

$$1 + \begin{matrix}\text{nominal}\\ \text{interest}\\ \text{rate}\end{matrix} = \left(1 + \begin{matrix}\text{real}\\ \text{interest}\\ \text{rate}\end{matrix}\right) * \left(1 + \begin{matrix}\text{anticipated}\\ \text{inflation}\\ \text{rate}\end{matrix}\right)$$

Valuation with inflation

- When valuing a project, two methods are correct:
  - Discount anticipated nominal cash flows at nominal interest rates.
  - Discount anticipated real cash flows at real interest rates.

Illustrating: Real cash flows vs nominal cash flows

	A	B	C	D	E	F	G
1	**CAPITAL BUDGETING FOR THE WIDGET MACHINE**						
2	Inflation rate	4.00%					
3	Widget price today	15.00					
4	Nominal discount rate	12.00%					
5	Equivalent real discount rate	7.69%	<-- =(1+B4)/(1+B2)-1				
6							
7	**Year**	**Widgets sold**	**Anticipated nominal widget price**	**Anticipated nominal cash flow**		**Anticipated real cash flow in year 0 dollars**	
8	0			-9,500.00		-9,500.00	
9	1	100	15.60	1,560.00	<-- =C9*B9	1,500.00	<-- =D9/(1+$B$2)^A9
10	2	125	16.22	2,028.00	<-- =C10*B10	1,875.00	<-- =D10/(1+$B$2)^A10
11	3	150	16.87	2,530.94		2,250.00	
12	4	160	17.55	2,807.66		2,400.00	
13	5	170	18.25	3,102.46		2,550.00	
14	6	200	18.98	3,795.96		3,000.00	
15							
16	**NPV calculations**			=$B$3*(1+$B$2)^A9			
17	Discounting nominal cash flows at nominal discount rates	778.93	<-- =NPV(B4,D9:D14)+D8				
18	Discounting real cash flows at real discount rates	778.93	<-- =NPV(B5,F9:F14)+F8				
19							
20	**IRR calculations**						
21	Nominal IRR	14.47%	<-- =IRR(D8:D14)				
22	Real IRR	10.06%	<-- =IRR(F8:F14)				
23	(1+nominal IRR)/(1+inflation)-1	10.06%	<-- =(1+B21)/(1+B2)-1				